
C65 Notepad

Manual

(for Version 1.0)

A simple text notepad for the C65



You need one of this options to use C65 notepad:

- C65
- MEGA 65
- Nexys A7 board with MEGA 65 bitstream
- Xemu emulator (xc65 or xmega65)

Because C65 notepad uses BASIC 10 commands which are only 
implemented in the latest ROM, you have to use at least the 
C65 ROM "911001.bin".

If you use a patched C65 ROM "911001_ae.bin" or higher, you 
can display german umlauts and the Euro symbol. You can find 
the patch "AE" with instructions here:

https://www.forum64.de/index.php?thread/101225-c65-rom-patch-
ae-with-german-umlauts

C65 notepad is for free. Use it and have fun.

If you like C65 notepad, a donation to the 
"Seehundstation Norddeich" (seehundstation-norddeich.de)
will be very welcome. You can use a donation with Paypal (a 
link for it can be found at the website) or a bank transfer:

Verein zur Erforschung und Erhaltung des Seehundes e.V.
IBAN: DE62 2836 1592 0007 7771 11
BIC: GENODEF1MAR
Bank: Raiffeisen Volksbank Fresena Norden

C65 Notepad was programmed 
by Guenther Reiter (mail: 65software@gmx.de)
in July 2020

1. What is C65 Notepad and what not?

C65 Notepad is ... 

... a simple text notepad (with four pages) to type in short 
text notes which could be saved on disc in the internal disc 
drive.

It could be used e.g. for notes about the programs on the 
disc (like readme docs etc.), a shopping list, your favorite 
cookie recipe, collect your ideas or whatever you want to 
take a short note of.

C65 Notepad is not ... 

... a wordprocessor to type your long loveletter to your 
sweetheart, your newest bestseller novel or your 
dissertation.

(Figure: Menu screen of C65 notepad)

2. Start C65 Notepad

To start C65 Notepad insert the program disc and type

RUN "C65NOTEPAD"

After start the program reads the file "NPAGES" from the 
disc, if it exists. Otherwise all four pages are empty.

The first page is shown and all your typing is in the first 
page. You can see the current page in the upper left corner 
of the screen. The current position of the cursor is shown 
next to the page display as <line, column>.

3. Typing notes

You can type your notes with the help of some special keys:

< INST > Insert one character and shift the right characters
in the line one position to the right.

< DEL > Delete one charachter an shift the right characters
in the line one position to the left.

< HOME > Set cursor home (into the upper left position 1,0).

< TAB > Set the cursor to the next tab position (every 8th 
column in the current line).

You can use the whole page for your text (24 lines and 80 
columns each). The current position of the cursor is shown in
the title bar as <line, column>.

4. Using the menu

With pressing the F1 key (or ESC key) you open the menu page.
In the background your current line was shown in yellow 
color. 

You have the following options with pressing the appropriate 
key:

< 1 > Go to page 1

< 2 > Go to page 2

< 3 > Go to page 3

< 4 > Go to page 4

< F1 > or < ESC > Go back to notepad

< H > Show the HELP page

< C > Clear the current page

< D > Delete the current line

< I > Insert one line and shift the lines below
one line down

< M > Start block marker

< V > Insert the content of clipboard to 
current position

< S > Save all four pages to disc

< B > Bsave current page for BLOAD to disc

< P > Export the current page as PETSCII text

< U > Export the current page as UFT-8 text

< X > Exit notepad without saving the pages

< Q > Save all pages and quit notepad

5. Description of the menu options

In the menu you can chose among the options as mentioned in 
the last chapter.

Below you can find the descriptions of the menu options who 
need some more words about.

5.1. Mark the begin of a block < M >

With pressing < M > in the menu you set the starting point of
a block to the current position of the cursor.

In the titlebar you can see a blinking message "<M> mark/<D> 
del" and the current cursor turns to yellow. 

You can move the cursor and the text between the starting 
point and the current positions turns to yellow background. 
So you can easily see which text section you have marked.

If you press < D >, the selected block is deleted. 

By pressing < M > you set the ending point of the block to 
the current cursor position. The selected text is copied into
the clipboard and you can see a "C" in the titlebar (for 
clipboard).

5.2. Insert the content of the clipboard < V >

By pressing < V > the content of the clipboard is insert at 
the current position.

You can insert the content as often as you like, because 
after inserting it the clipboard will be not deleted.

Only if you mark a new text and copy it into to clipboard the
old text will be deleted.

You can copy the text to every page of C65 notepade and not 
only to that page from which you get it.

5.3. Save pages to disc < S >

By pressing < S > you can save all four pages to disc. The 
file is named "NPAGES" and uses 33 blocks on disc.

You can not change the filename. If you save the pages an 
existing file "NPAGES" will be deleted.

If you want to backup a saved file, you can use the BASIC 
command COPY to copy the file to a file with another 
filename:

COPY "NPAGES" TO "MYFILENAME"

5.4. Bsave page to disc < B >

Pressing < B > saves the current page to disc. The file is 
named "BPAGE1", "BPAGE2", "BPAGE3" or "BPAGE4" depending on 
your current page number and uses 8 blocks on disc.

You can not change the filename. If you save the page an 
existing file "BPAGES<x>" will be deleted.

If you want to backup a saved file, you can use the BASIC 
command COPY to copy the file to a file with another filename
like: 

COPY "BPAGES1" TO "MYBPAGE1"

The Bsave option saves the page only for external use. The 
created file is not used for loading the text into C65 
notepad again.

You can show the saved page with the BASIC command:

BLOAD "BPAGE1", P 2048, B 0
(this shows the saved page 1)

Or you can use the program "README" which is stored on the 
program disc. For further details see below.

5.5. Export page as PETSCII text < P >

Pressing < P > exports the text of the current page a 
sequential text file in PETSCII format to disc.

The filename depends on the current page and is for instance 
for page 1 "BTEXT1PETSCII".

You can not change the filename. If you save the page an 
existing file "BTEXT<x>PETSCII" will be deleted.

If you want to backup a saved file, you can use the BASIC 
command COPY to copy the file to a file with another filename
like: 

COPY "BTEXT1PETSCII" TO "MYBTEXT1PETSCII"

You can show the saved PETSCII text with the BASIC command:

TYPE "BTEXT1PETSCII"
(shows the saved PETSCII text from paged 1)

5.6. Export page as UTF-8 text < U >

Pressing < U > exports the text of the current page a 
sequential text file in UTF-8 format to disc. This format 
allows to use the text in nearly all current computer 
systems.

The filename depends on the current page and is for instance 
for page 1 "BTEXT1UTF8".

You can not change the filename. If you save the page an 
existing file "BTEXT<x>UTF8" will be deleted.

If you want to backup a saved file, you can use the BASIC 
command COPY to copy the file to a file with another filename
like: 

COPY "BTEXT1UTF8" TO "MYBTEXT1UTF8"

To use the UTF-8 formated text from the disc, you need the 
help of an external program. I recommend for that purpose 
"DirMaster/Style" (style64.org/dirmaster) which can open a 
disc image file (*.D81), display the directory of the disc 
and allow to export a file to the harddrive of your current 
computer.

After starting DirMaster you have to open your disc image in 
which your UTF-8 formated text is saved. DirMaster shows you 
the directory of this disc.

(Figure: Open and choose "Export" with DirMaster)

Select the file with a right mouse click and chose the option
"Export". Be aware that you have to select "PETSCII (*.seq)" 
as the file type before your save the file. 

(Figure: Export file with file type "PESTSCII (*.seq)")

Finally rename the saved file into a file with the file 
extension for text files (*.txt). 

You can open and display the file correctly with almost every
modern text editor. 

5.7. Exit C65 notepad without saving the pages < X >

With pressing < X > you leave C65 notepad. Be aware, that the
pages are not saved before you exit. Any text changes since 
your last save will be lost.

5.8. Save pages and quit C65 notepad < Q >

By pressing < Q > all pages are saved to disc with filename 
"NPAGES". An existing file with that name will be deleted 
before saving. After that the C65 notpad will be closed.

6. Some hints using C65 notepad

6.1. German umlauts

If you use a patched C65 ROM (at least patch "AE"), you can 
type and display german umlauts (Ä,Ö,Ü,ä,ö,ü,ß) and the Euro 
symbol "€". 

To write e.g. an "ä" you have to press the Commodore key (C=)
together with the "A" key. The uppercase umlauts lay one row 
below and half right under the appropriate lowercase umlaut.

< C= > + < Z > Ä

< C= > + < L > Ö

< C= > + < J > Ü

< C= > + < A > ä

< C= > + < O > ö

< C= > + < U > ü

< C= > + < S > ß

< C= > + < E > €

For instructions and patch "AE" download see here: 

https://www.forum64.de/index.php?thread/101225-c65-rom-patch-
ae-with-german-umlauts

6.2. Use C65 notepad for creating readme files

To create a "readme file" for a disc of your software, you 
can easily use C65 notepad with the additional program 
"README" (which is included on the program disc).

First write your instructions or whatever you like text with 
C65 notepad. Start with page 1 and if you need more space go 
on with page 2, page 3 and page 4.

Save every page as "Bsave" (using the < B >) command to disc.

This way creates the file "BPAGE1" for the text of page 1 and
appropiated files for the other pages.

Starting the program "README" (e.g. with RUN "README"), the 
first page ("BPAGE1") will be shown on screen and the program
waits for a keypress.

If a second page file exists on disc ("BPAGE2") the program 
will show then this text ... and so on. 

After the fourth page or if a further page file doesn't 
exists, the program clears the screen and stop.

So this is an easy way to produce explaining texts you can 
deliver together with your discs.

7. Sourcecode

"C65 notepad" and "Readme" are programmed with BASIC 10.

In the following pages you can find the sourcecode of 
"Readme" and "C65 notepad".

10 rem readme for c65 notepad

20 rem

30 rem shows you the four screens (bpage...) saved by c65 

notepad

40 rem

50 rem programmed by guenther reiter, 2020/07

60 rem

70 rem

80 scnclr

90 trap 180 : rem if page file not found, jump to end of 

program

100 bload"bpage1",p 2048, b 0 : rem load "bpage1" into 

screen

110 getkey k$ : rem wait for a keypress

120 bload"bpage2",p 2048, b 0 : rem load "bpage2" into 

screen

130 getkey k$ : rem wait for a keypress

140 bload"bpage3",p 2048, b 0 : rem load "bpage3" into 

screen

150 getkey k$ : rem wait for a keypress

160 bload"bpage4",p 2048, b 0 : rem load "bpage4" into 

screen

170 getkey k$ : rem wait for a keypress

180 scnclr

190 end

ready.

10 rem c65 notepad v1.0

15 rem 2020/07 by guenther reiter

16 rem

20 dim u$(256):gosub 7000:rem set utf-8 export signs

21 key 1,chr$(27)

22 bk=4:cl=0

23 bank bk:poke 2048,65:p=peek(2048)

24 if (p=65) then goto 27

25 bk=bk+1:if bk=255 then goto 27

26 goto 23

27 rem set temp bank to correct number

30 rem trap 900 : rem on any error goto 900

35 border 6:background 6:foreground 1:scnclr

36 t$="sr                                 [F1 = Menu]     

*** C65 NOTEPAD V1.0 ***"

37 print chr$(14):rem change to lowercase charset

38 print chr$(11):rem prevent switching to uppercase 

charset

40 pg=1:cx=0:cy=1:bank 0:nt=0

41 trap 45

42 bload "npages",p 2048, b(bk): rem load pages from disc 

into ram (bank bk)

44 goto 50

45 dma 3,2048*4,32,0,2048,bk : rem clear pages in bank bk

46 trap 900

50 dma 0,2000,2048*1,bk,2048,0 : rem display page 1

80 print t$;:poke 2127,160

90 print "sr Page"+str$(pg)+" ["+right$(str$

(cy),2)+","+right$(str$(cx),2)+"] ’";

95 if cl>0  then cursor 49,0:print " C ";

100 p=2048+80*cy+cx:pp=peek(p):if (pp<128) then 

pp=pp+128:goto 102

101 pp=pp-128

102 poke p,pp:bank 1:poke p+61440,8:bank 0:rem set cursor 

color (red)

105 sk=1

110 getkey k$

112 if nt>0 then cursor 18,0:print "r               

’";:nt=0

115 pp=peek(p)

116 if pp>128 then pp=pp-128:goto 119

118 pp=pp+128

119 poke p,pp:bank 1:poke p+61440,1 :bank 0

120 k=asc(k$)

129 if k=27  then gosub 2100 : rem escape

130 if k=148 then gosub 3800 : rem insert

131 if k=29  then gosub 1000 : rem cursor right

132 if k=157 then gosub 1100 : rem cursor left

133 if k=17  then gosub 1200 : rem cursor down

134 if k=145 then gosub 1300 : rem cursor up

135 if k=13  then gosub 1400 : rem return

136 if k=20  then gosub 1500 : rem delete (back)

143 if k=19 then gosub 3500 : rem home

144 if k=9 then gosub 3600 : rem tab

145 if sk=0 then goto 190

149 kp=k

150 if (k>-1)  and (k<32) then kp=k+128:goto 175

151 if (k>63)  and (k<96) then kp=k-64:goto 175

152 if (k>95)  and (k<160) then kp=k+64:goto 175

153 if (k>160) and (k<192) then kp=k-64:goto 175

154 if (k>191) and (k<224) then kp=k-128

175 p=2048+80*cy+cx:poke p,kp

180 cx=cx+1

185 if cx=80 then cx=0:cy=cy+1

187 if cy=25 then cy=1

190 goto 90

200 end

900 rem trap

910 scnclr

930 print "“r *** C65 NOTEPAD V1.0 *** ’":print chr$(12)

950 print "exit."

960 rem dma 3,14000,0,0,8192,0

990 end

999 rem

1000 rem cursor right

1010 cx=cx+1:sk=0

1020 if cx=80 then cy=cy+1:cx=0

1030 if cy=25 then cy=1

1040 return

1100 rem cursor left

1110 cx=cx-1:sk=0

1120 if cx=-1 then cy=cy-1:cx=79

1130 if cy=0  then cy=24

1140 return

1200 rem cursor down

1210 cy=cy+1:sk=0

1220 if cy=25 then cy=1

1240 return

1300 rem cursor up

1310 cy=cy-1:sk=0

1320 if cy=0 then cy=24

1340 return

1400 rem return

1410 cx=0:cy=cy+1:sk=0

1420 if cy=25 then cy=1

1430 return

1500 rem delete (back)

1510 cx=cx-1:k=32:sk=0

1520 if cx<0 then cx=0:goto 1590

1540 dma 0,2000,2048,0,2048*6,bk: rem store page 1 at temp

(bank bk)

1550 dma 0,80-cx-1,2048*6+80*cy+cx+1,bk,2048+80*cy+cx  ,0

1560 poke 2048+80*cy+79,32 : rem clear last char in line 

cy

1590 return

1600 rem page 1

1610 dma 0,2000,2048,0,2048*pg,bk

1620 dma 0,2000,2048*1,bk,2048,0

1625 print t$;:poke 2127,160

1630 sk=0:pg=1:cx=0:cy=1

1640 return

1700 rem page 2

1710 dma 0,2000,2048,0,2048*pg,bk

1720 dma 0,2000,2048*2,bk,2048,0

1725 print t$;:poke 2127,160

1730 sk=0:pg=2:cx=0:cy=1

1740 return

1800 rem page 3

1810 dma 0,2000,2048,0,2048*pg,bk

1820 dma 0,2000,2048*3,bk,2048,0

1825 print t$;:poke 2127,160

1830 sk=0:pg=3:cx=0:cy=1

1840 return

1900 rem page 4

1910 dma 0,2000,2048,0,2048*pg,bk

1920 dma 0,2000,2048*4,bk,2048,0

1925 print t$;:poke 2127,160

1930 sk=0:pg=4:cx=0:cy=1

1940 return

2000 rem escape

2005 dma 0,2000,2048,0,2048*pg,bk

2010 scratch "npages"

2015 print "“r *** C65 NOTEPAD V1.0 *** 

’":print:print"saving pages to disc ..."

2020 bsave "npages",p 2048 to p (2048*5), b(bk)

2030 print:print "done.":print chr$(12)

2035 rem dma 3,14000,0,0,8192,0

2040 end

2050 rem exit

2055 print "“r *** C65 NOTEPAD V1.0 *** ’":print chr$(12)

2060 print "exit."

2070 rem  dma 3,14000,0,0,8192,0

2090 end

2100 rem menu

2102 dma 0,2000,2048,0,2048*pg,bk

2105 sk=0:wx=27

2106 bw$=chr$(144)+"r ’"+chr$(5)

2107 dma 3,80, 7+32 ,1,dec("f800")+80*cy,1:rem set color 

of current line

2110 cursor wx, 2:print "r                         ’";

2120 cursor wx, 3:print "r      NOTEPAD MENU       ’";bw$;

2130 cursor wx, 4:print "r                         ’";bw$;

2140 cursor wx, 5:print "r [1-4] Go to page 1-4    ’";bw$;

2142 cursor wx, 6:print "r [F1]  Back to notepad   ’";bw$;

2150 cursor wx, 7:print "r [H]   Show help page    ’";bw$;

2155 cursor wx, 8:print "r                         ’";bw$;

2160 cursor wx, 9:print "r [C] Clear current page  ’";bw$;

2180 cursor wx,10:print "r [D] Delete current line ’";bw$;

2200 cursor wx,11:print "r [I] Insert one line     ’";bw$;

2210 cursor wx,12:print "r [M] Start block marker  ’";bw$;

2215 cursor wx,13:print "r [V] Insert clipboard    ’";bw$;

2220 cursor wx,14:print "r [S] Save pages to disc  ’";bw$;

2225 cursor wx,15:print "r [B] Bsave page to disc  ’";bw$;

2230 cursor wx,16:print "r [P] Export as PETSCII   ’";bw$;

2235 cursor wx,17:print "r [U] Export as UFT-8     ’";bw$;

2240 cursor wx,18:print "r                         ’";bw$;

2320 cursor wx,19:print "r [X] Exit without saving ’";bw$;

2340 cursor wx,20:print "r [Q] Save pages and quit ’";bw$;

2350 cursor wx,21:print "r                         ’";bw$;

2360 cursor wx+1,22:print chr$(144);"r                    

’";chr$(5);

2460 getkey k$

2470 dma 0,2000,2048*pg,bk,2048,0

2475 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

2480 rem

2490 if (k$="q") or (k$="Q") then goto 2000

2500 if (k$="x") or (k$="X") then goto 2050

2510 if (k$="c") or (k$="C") then goto 3100

2520 if (k$="s") or (k$="S") then goto 3200

2530 if (k$="d") or (k$="D") then goto 3300

2540 if (k$="i") or (k$="I") then goto 3400

2545 if (k$="v") or (k$="V") then goto 5800

2550 if (k$="b") or (k$="B") then goto 3700

2552 if (k$="m") or (k$="M") then goto 4500

2553 if (k$="p") or (k$="P") then goto 3900

2555 if (k$="u") or (k$="U") then goto 5900

2557 if (k$="h") or (k$="H") then gosub 4200

2560 if (k$="1") then gosub 1600

2565 if (k$="2") then gosub 1700

2570 if (k$="3") then gosub 1800

2575 if (k$="4") then gosub 1900

2599 return

3000 rem help

3010 sk=0

3100 rem clear page

3120 dma 3,1920,32,0,2048+80,0 : rem clear page

3190 goto 90

3200 rem save pages

3205 dma 0,2000,2048,0,2048*pg,bk

3210 scratch "npages"

3215 cursor 18,0:print " saving ... ";

3216 dma 3,12,16+1,0,dec("f800")+18,1 : rem blink

3220 bsave "npages",p 2048 to p (2048*5), b(bk)

3230 cursor 18,0:print " pages saved ";:nt=1

3290 goto 90

3300 rem delete current line

3310 sk=0:sl=2048+80*cy

3340 dma 0,2000,2048,0,2048*6,bk: rem store page 1 at temp

(bank bk)

3350 dma 0,2000-(cy*80),2048*6+80*(cy+1),bk,sl,0

3360 dma 3,80,32,0,2048+80*24,0 : rem clear line 25

3390 goto 90

3400 rem insert one line

3410 sk=0:sl=2048+80*cy

3420 rem   poke sl,65

3430 rem dma 3,80,65,0,sl,0 : rem clear line

3440 dma 0,2000,2048,0,2048*6,bk: rem store page 1 at temp

(bank bk)

3450 dma 0,2000-(cy*80),2048*6+80*(cy-1),bk,sl,0: rem 

store page 1 at temp

3460 dma 3,80,32,0,sl,0 : rem clear current line

3490 goto 90

3500 rem home

3510 cx=0:cy=1:sk=0

3590 return

3600 rem tab

3610 sk=0:cx=(int(cx/8)+1)*8

3620 if cx>79 then cx=79

3690 return

3700 rem bsave page to disc

3710 sk=0:f$="bpage"+right$(str$(pg),1)

3715 dma 0,2000,2048,0,2048*6,bk: rem store page 1 at temp

(bank bk)

3717 dma 3,80,32,0,2048,0 : rem clear line 0

3720 scratch(f$)

3730 bsave(f$),p 2048 to p (2048+80*25),b 0

3740 dma 0,2000,2048*6,bk,2048,0: rem restore page 1 from 

temp (bank bk)

3790 goto 90

3800 rem insert

3810 sk=0

3820 dma 0,2000,2048,0,2048*6,bk: rem store page 1 at temp

(bank bk)

3830 if cx<79 then dma 0,80-cx-

1,2048*6+80*cy+cx,bk,2048+80*cy+cx+1,0

3840 poke 2048+80*cy+cx,32

3890 return

3900 rem export text as petscii

3910 sk=0

3915 fs=dec("f800")-2048*6

3920 dma 0,2000,2048,0,2048*6,bk: rem store page pg at 

temp (bank bk)

3930 cursor 18,0:print " export PETSCII ";

3935 dma 3,15,16+1,0,dec("f800")+18,1 : rem blink

3936 f$="btext"+right$(str$(pg),1)+"petscii"

3938 scratch (f$)

3940 dopen#1,(f$),w

3950 bank bk

3960 rem    for l=2048*6+80*1 to 2048*6+80*25-1 step 80

3962 for l=2048*6+80*1 to 2048*6+80*25-1 step 80

3970 e$=""

3980 for p=l to (l+79)

3990 t=peek(p)

3995 bank 1:poke fs+p,7+32:bank(bk)

4000 if (t>-1) and (t<32) then c=t+64:goto 4070

4010 if (t>31)and(t<64) then c=t:goto 4070

4020 if (t>63) and (t<96) then c=t+128:goto 4070

4030 if (t>95) and (t<128) then c=t+64:goto 4070

4040 if (t>127) and (t<160) then c=t-128:goto 4070

4050 if (t>159) and (t<224) then c=t-64:goto 4070

4060 if (t>223) then c=t

4070 e$=e$+chr$(c)

4080 next p

4090 print#1,e$

4100 next l

4110 dclose#1

4122 cursor 0,0:print t$;:poke 2127,160:nt=0

4125 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

4130 bank 0

4190 goto 90

4200 rem help

4202 dma 0,2000,2048,0,2048*pg,bk

4205 sk=0:wx=25

4206 bw$=chr$(144)+"r ’"+chr$(5)

4210 cursor wx, 2:print "r                             ’";

4220 cursor wx, 3:print "r    NOTEPAD HELP - KEYS      

’";bw$;

4237 cursor wx, 4:print "r                             

’";bw$;

4240 cursor wx, 5:print "r  [INST] insert one char     

’";bw$;

4250 cursor wx, 6:print "r  [DEL]  delete one char     

’";bw$;

4260 cursor wx, 7:print "r  [HOME] set cursor home     

’";bw$;

4265 cursor wx, 8:print "r  [TAB]  cursor to next tab  

’";bw$;

4267 cursor wx, 9:print "r                             

’";bw$;

4268 cursor wx,10:print "r     GERMAN SPECIAL KEYS     

’";bw$;

4269 cursor wx,11:print "r                             

’";bw$;

4270 cursor wx,12:print "r  [C= + Z] Ae  [C= + A] ae   

’";bw$;

4271 cursor wx,13:print "r  [C= + L] Oe  [C= + O] oe   

’";bw$;

4272 cursor wx,14:print "r  [C= + J] Ue  [C= + U] ue   

’";bw$;

4273 cursor wx,15:print "r  [C= + S] ss  [C= + E] EUR  

’";bw$;

4277 cursor wx,16:print "r                             

’";bw$;

4280 cursor wx,17:print "r                             

’";bw$;

4300 cursor wx,18:print "r  C65 NOTEPAD - Version 1.0  

’";bw$;

4305 cursor wx,19:print "r     programmed 2020/07      

’";bw$;

4310 cursor wx,20:print "r     by Guenther Reiter      

’";bw$;

4315 cursor wx,21:print "r                             

’";bw$;

4320 cursor wx,22:print "r        used bank:";right$(str$

(bk),2);"         ’";bw$;

4325 cursor wx,23:print "r                             

’";bw$;

4350 cursor wx+1,24:print chr$(144);"r                    

’";chr$(5);

4360 getkey k$

4370 dma 0,2000,2048*pg,bk,2048,0

4375 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

4380 rem

4499 return

4500 rem start marker block

4510 mx=cx  :my=cy

5000 rem

5050 l=(80*cy+cx)-(80*my+mx):if l<1 then l=1:cx=mx+1:cy=my

5053 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

5055 dma 3,l, 7+32 ,1,dec("f800")+80*my+mx,1:rem set 

marker block color

5060 cursor 16,0:print "[M] mark/[D] del";:nt=1

5070 dma 3,16,16+1,0,dec("f800")+16,1 : rem blink

5100 getkey k$

5110 k=asc(k$)

5131 if k=29  then gosub 5200 : rem cursor right

5132 if k=157 then gosub 5300 : rem cursor left

5133 if k=17  then gosub 5400 : rem cursor down

5134 if k=145 then gosub 5500 : rem cursor up

5140 if k=27  then gosub 5600 : rem escape

5150 if k=77 or k=205 then goto 5700 : rem "m" mark end

5160 if k=68 or k=196 then goto 7200 : rem "d" delete 

block

5190 goto 5050

5200 rem marker - cursor right

5210 cx=cx+1

5220 if cx>79 then cx=0:cy=cy+1

5230 if cy>24 then cy=24

5290 return

5300 rem marker - cursor left

5310 cx=cx-1

5320 if cx<0  then cx=79:cy=cy-1

5330 if cy<1  then cy=1

5390 return

5400 rem marker - cursor down

5410 cy=cy+1

5430 if cy>24 then cy=24

5490 return

5500 rem marker - cursor up

5510 cy=cy-1

5530 if cy<1  then cy=1

5590 return

5600 rem marker - escape

5610 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

5620 if nt>0 then cursor 18,0:print "r              

’";:nt=0

5690 goto 90

5700 rem marker - copy

5710 ca=2048+80*my+mx:rem start address of block

5720 cb=2048+80*cy+cx:rem   end address of block

5730 cl=cb-ca : rem length of block in clipboard

5740 dma 0,cl  ,ca  ,0,2048*5,bk: rem store block in bank 

bk

5760 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

5770 if nt>0 then cursor 16,0:print "r                

’";:nt=0

5790 goto 90

5800 rem insert clipboard

5805 if cl=0 then goto 5890

5810 ca=2048+80*cy+cx : rem start address

5820 cb=ca+cl : rem end address

5830 ce=2048+80*25

5840 if cb>ce then cl=ce-ca

5850 dma 0,cl,2048*5,bk,ca  ,0 : rem get stored block from

bank bk

5890 goto 90

5900 rem export text as utf-8

5910 sk=0

5915 fs=dec("f800")-2048*6

5920 dma 0,2000,2048,0,2048*6,bk: rem store page pg at 

temp (bank bk)

5930 cursor 18,0:print " export UTF-8 ";

5935 dma 3,13,16+1,0,dec("f800")+18,1 : rem blink

5936 f$="btext"+right$(str$(pg),1)+"utf8"

5938 scratch (f$)

5940 dopen#1,(f$),w

5950 bank bk

5955 cursor 0,5

5960 for l=2048*6+80*1 to 2048*6+80*25-1 step 80

5962 rem for l=2048*6+80*1 to 2048*6+80*5 -1 step 80

5970 e$=""

5980 for p=l to (l+79)

5990 t=peek(p)

5995 bank 1:poke fs+p,7+32:bank(bk)

6070 e$=e$+u$(t)

6080 next p

6090 print#1,e$

6100 next l

6110 dclose#1

6120 rem  cursor 18,0:print "   finished  ";:nt=1

6122 cursor 0,0:print t$;:poke 2127,160:nt=0

6125 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

6130 bank 0

6190 goto 90

7000 rem datas for utf-8 export

7010 for i=0 to 255:u$(i)=chr$(32):next i: rem set all 

signs to "space" default

7011 u$(109)=chr$(195)+chr$(132): rem uppercase ae

7012 u$(118)=chr$(195)+chr$(150): rem uppercase oe

7013 u$(117)=chr$(195)+chr$(156): rem uppercase ue

7015 for i=65 to 90:u$(i)=chr$(i):next i:rem set uppercase

letters a-z

7020 for i=1 to 26:u$(i)=chr$(i+96):next i:rem set 

lowercase letters a-z

7021 u$(112)=chr$(195)+chr$(164): rem lowercase ae

7022 u$(121)=chr$(195)+chr$(182): rem lowercase oe

7023 u$(120)=chr$(195)+chr$(188): rem lowercase ue

7024 u$(110)=chr$(195)+chr$(159): rem sz

7030 for i=48 to 57:u$(i)=chr$(i):next i:rem set 0-9

7035 for i=33 to 41:u$(i)=chr$(i):next i:rem set shift 1-9

7040 u$(45)=chr$(45):rem -

7042 u$(93)=chr$(124):rem vertical line

7044 u$(61)=chr$(61):rem =

7046 u$(42)=chr$(42):rem *

7048 u$(64)=chr$(45):rem hor line

7050 u$(30)=chr$(94):rem ^

7052 u$(94)=chr$(35):rem checked block

7054 u$(58)=chr$(58):rem :

7056 u$(27)=chr$(91):rem [

7058 u$(29)=chr$(93):rem ]

7060 u$(59)=chr$(59):rem ;

7062 u$(44)=chr$(44):rem ,

7064 u$(46)=chr$(46):rem .

7066 u$(60)=chr$(60):rem <

7068 u$(62)=chr$(62):rem >

7070 u$(47)=chr$(47):rem /

7072 u$(63)=chr$(63):rem ?

7090 return

7200 rem marker - delete

7210 ca=2048+80*my+mx:rem start address of block

7220 cb=2048+80*cy+cx:rem   end address of block

7230 cl=cb-ca : rem length of block in clipboard

7240 dma 3,cl,32,0,ca,0 : rem clear marked block

7260 dma 3,2000,1,1,dec("f800"),1:rem set all color cells 

to white (1)

7270 if nt>0 then cursor 16,0:print "r                

’";:nt=0

7290 goto 90

ready.

